Formalisation of Hall's Theorem for Countable Infinite Graphs*

Fabián Fernando Serrano Suárez[1], Mauricio Ayala-Rincón[2], and Thaynara Arielly de Lima[3]

[1] Universidad Nacional de Colombia - Sede Manizales ffserranos@unal.edu.co
[2] Universidade de Brasília, D.F. Brazil ayala@unb.br
[3] Universidade Federal de Goiás, Goiânia Brazil thaynaradelima@ufg.br

Abstract. This work presents two formalisations in Isabelle/HOL of the extension of Hall's marriage theorem for finite graphs to countable infinite graphs. The proofs use a formalisation of the authors' countable set-theoretical version of Hall's theorem, which was proved using a formalisation in Isabelle/HOL of the compactness theorem for propositional logic by dealing with finite families of sets through the well-known marriage-condition characterisation. The first formalisation focuses on maintaining specifications and proofs as closely as possible to textbook proofs. The second one states the theorem directly in terms of the existence of perfect matchings over finite and infinite graphs, profiting from the conciseness of Isabelle/HOL locales' technology. The development contributes to mechanising countable infinite versions of properties equivalent to Hall's marriage theorem in contexts other than set theory.

Keywords: Interactive Theorem Proving · Graph Theory · Set Theory · Combinatorics · Automated deduction

1 Introduction

Hall's marriage theorem is a landmark result established primarily by Philip Hall [17], and it is equivalent to several other significant theorems in combinatorics and graph theory (cf. [8], [9], [28]), namely: Menger's theorem (1929), König's minimax theorem (1931), König–Egerváry theorem (1931), Dilworth's theorem (1950), Max Flow-Min Cut theorem (related to the well-known Ford-Fulkerson

* Research funded by grants: CNPq Universal 409003/21-2, and Productivity 313290/2021-0, FAPDF DE 00193-00000229/2021-21, FAPEG 202310267000223, and CAPES PrInt under financial code 001. The second and third authors are grateful to the Hausdorff Institute für Mathematik (Universität Bonn) for hosting them as part of the trimester "Prospects of formal Mathematics", funded by the Deutsche Forschungsgemeinschaft (DFG) under Germany Excellence Strategy – EXC-2047/1 – 390685813.
A preliminary version of this work was presented at the 24th Int. Conf. on Logic for Programming Artificial Intelligence and Reasoning (available as preprint [33]).

algorithm), among others. Consequently, any mechanisation of Hall's theorem allows one to prove any of those equivalent results formally.

Two well-known versions of Hall's theorem exist, one for *finite* families of finite sets and another for *finite* graphs. The proofs of any previously cited equivalences can be more adapted to a specific version of Hall's Theorem, either the set-theoretical or the graph-theoretical version. For example, König–Egerváry theorem states that the minimum cover in a finite bipartite graph has the same cardinality as a maximum matching. Thus, if we assume Hall's theorem for finite graphs, one possible way to infer König–Egerváry theorem will consist of building a reduction from the latter to the former. Considering the nature of König–Egerváry theorem, it is clear that the graph-theoretical version of Hall's theorem is more appropriate than the set-theoretical version to establish the equivalence between these theorems.

Although we referred to the *finite* versions of the mentioned results in the previous paragraphs, we point out that extensions to *infinite* sets and graphs are of primary interest [2].

Mechanisations such as those presented in this work aim to pave the way to develop formalisations of *infinite* versions of some theorems in combinatorics related to Hall's Theorem. For example, the authors formalised the set-theoretical version of Hall's Theorem for a countable (infinite) collection of finite subsets $\{S_i\}_{i \in I}$ of a set S [32]. Such a development applied a formalisation of the compactness theorem for propositional logic, developed by Serrano in [31], and Jiang and Nipkow's formalisation for the finite case of the set-theoretical version of Hall's theorem [21].

As main results, this work discusses how applying authors' development in [32], the *infinite* graph-theoretical version of Hall's theorem is mechanised in Isabelle/HOL. The result applies to a general class of *infinite* bipartite graphs with finite neighbourhoods regarding one of the sets of vertices of the vertex bipartition. Additionally, a second succinct formalisation of the same result that uses Isabelle locales is also discussed. The formalisations are of practical interest since they can be used to establish the mechanisation of other combinatorial results, as the previous ones discussed, over *infinite* sets and graphs.

Interestingly, other combinatorial well-known results equivalent to Hall's theorem in the finite case are not straightforwardly equivalent in the infinite case; for instance, the infinite version of König-Egerváry theorem that as reported in [2] cannot be inferred from the compactness theorem.

Organisation. Section 2 discusses Hall's marriage theorem for finite and infinite countable sets and graphs and explains the equivalence between the versions for graphs and sets. Then, Section 3 presents the two formalisations in Isabelle/HOL of the graph-theoretical version of Hall's theorem for countable graphs. Section 4 discusses related work before concluding in Section 5. The paper includes links to the formalisation highlighted by the symbol ↗.

2 Hall's Theorem for sets and graphs

2.1 Finite and infinite versions of Hall's theorem

Hall's theorem for sets establishes that a finite family $\{S_i\}_{i\in I}$ of finite sets not necessarily disjoint, of elements in a set S, has a system of distinct representatives (SDR) if and only if the so-called *marriage condition* holds. The marriage condition states that:

$$\text{For any } J \subseteq I, |J| \leq |\bigcup_{j\in J} S_j|$$

Above, an SDR for the family $\{S_i\}_{i\in I}$ is understood as a subset of elements of S that contains exactly an element for each set in the family. This can be formalised as an injective function $f : I \to S$, such that $f(i) \in S_i$, for $i \in I$.

Definition 1 (SDR). *Let S be an arbitrary set and $\{S_i\}_{i\in I}$ a collection of not necessarily distinct subsets of S with indices in the set I. An injective function $f : I \to \bigcup_{i\in I} S_i$ is an SDR for $\{S_i\}_{i\in I}$ if for all $i \in I$, $f(i) \in S_i$.*

Using the compactness theorem, a proof of a countable infinite version of this theorem was formalised in Isabelle/HOL [32]. The infinite version states that a countable family of finite sets, indexed by a set I, has a set of distinct representatives if and only if Hall's *marriage condition* below holds:

$$\text{For any } J \subseteq I, J \text{ finite}, |J| \leq |\bigcup_{j\in J} S_j|$$

Hall's theorem for finite graphs states that in a bipartite graph $G = \langle X, Y, E\rangle$, (where $E \subseteq X \times Y$,) there is a perfect matching covering X if and only if $|J| \leq |N(J)|$ for all $J \subseteq X$. Here, for $x \in X \cup Y$, the neighbourhood of x is the set of vertices $N(x) = \{y \mid (x, y) \in E, \text{ or } (y, x) \in E\}$. N is extended straightforwardly to sets.

Definition 2 (Directed bipartite digraph and perfect matching). *Let X and Y be nonempty sets. The triple $G = \langle X, Y, E\rangle$ is a directed bipartite digraph if and only if the following conditions hold.*

1. $X \cap Y = \emptyset$, and 2. $E \subseteq (X \times Y)$.

A subset of arcs $E' \subseteq E$ is a perfect matching of $G = \langle X, Y, E\rangle$ if and only if

1. $X = \{x \mid (x, y) \in E'\}$, and 2. E' is an injective relation.

The infinite version of Hall's theorem for graphs states that in a countable bipartite graph $G = \langle X, Y, E\rangle$, where for all $x \in X$, $N(x)$ is finite, there is a perfect matching covering X if and only if $|J| \leq |N(J)|$ for all J finite, $J \subseteq X$. It may be directly graph-theoretically stated as "there exists a perfect matching if and only if for any finite subgraph there is a perfect matching." Indeed, the graph and each of their subgraphs translate into a family of indexed sets, $\{N(j)\}_{j\in J}$, and the system of distinct representatives allows the construction of a perfect matching consisting of the set of edges $\{(j, y) \mid j \in J, \text{ where } y \text{ is the representative of } j\}$.

Notice that for the infinite version of this theorem, the finiteness of $N(x)$ cannot be relaxed; in fact, the graph $G = \langle \mathbb{N}, \mathbb{N}^+, \{(0,i) \mid i \in \mathbb{N}^+\} \bigcup \{(i,i) | i \in \mathbb{N}^+\}\rangle$ is an easy counterexample. In G, the sets of vertices $\mathbb{N}$ and $\mathbb{N}^+$ are seen as different copies of natural numbers.

The formalisation of Hall's Theorem for countable families in [32] uses Nipkow's formalisation of Hall's theorem for finite families of sets [20] and Serrano's formalisation of the compactness theorem for propositional logic [31].

2.2 Countable versions of Hall's theorem for sets and graphs

The equivalence between countable versions of this theorem for sets and graphs is clear intuitively.

On the one side, a countable bipartite graph $G = \langle X, Y, E\rangle$ gives a countable family of neighbourhoods $\{N(x)\}_{x\in X}$, which are finite sets under the constraint that neighbourhoods of vertices in X are finite. If M is a perfect matching of G, thus one builds an SDR by considering the injective function $f : X \to Y$ such that, for each $x \in X$, $f(x) = y$, where $(x, y) \in M$.

On the other side, if one has a countable family of finite sets $\{S_i\}_{i\in I}$ satisfying the marriage condition, then there exists a distinct set of representatives for $\{S_i\}_{i\in I}$, given by f. We consider the countable bipartite graph built as $G = \langle I, \bigcup_{i\in I} S_i, E\rangle$, where $E = \{(i, y) \mid i \in I, y \in S_i\}$. Since the sets in the countable family of sets $\{S_i\}_{i\in I}$ are finite the set of neighbourhoods in G, for each $i \in I$, $N(i)$, is finite; indeed, $|S_i| = |N(i)|$. Since f is injective, the perfect matching covering I is given by the set of arcs $M = \{(i, f(i)) \mid i \in I\}$.

3 Formalisation of Hall's Theorem for Graphs

Initially, we discuss how infinite families of sets and infinite bipartite graphs are specified. Afterwards, we explain how the proof of correction of the specialised construction of an SDR from a perfect matching over an infinite directed bipartite graph is used to conclude the infinite graph-theoretical version of Hall's theorem. Finally, a formalisation using Isabelle locales is presented.

3.1 Formalising relations between sets and graphs

The formalisation is constructive, and its kernel is the transformations of indexed infinite families of sets to and from directed bipartite digraphs. One of the vital features of our formalisation is how we build a *system of distinct representatives* (SDR) for a family of sets from a perfect matching over arbitrary directed bipartite graphs. Such transformations are more general than those discussed in the previous section since neither the family of sets need to be countable nor the sets in the family must be restricted to finite sets. Thus, the bipartite graph may also be non-countable, and the neighbourhoods of the vertices do not need to be finite. Theorems 1 and 2 present the reductions from a problem to another one and state that from the existence of a perfect matching, the resulting transformation is an indexed family of sets that has an SDR, and vice-versa.

Theorem 1 (SDR associated to a directed bipartite digraph). *Let $G = \langle X, Y, E \rangle$ be a directed bipartite digraph.*

The collection of sets associated to G is built as $\{V_i\}_{i \in I}$, where $I = X$, and for all $i \in I$, $V_i = \{y \mid (i, y) \in E\}$.

Therefore, if E' is a perfect matching of G, the function $R : I \to \bigcup_{i \in I} V_i$, defined as $R(i) = y$, where y is the unique element in V_i such that $(i, y) \in E'$, is an SDR of $\{V_i\}_{i \in I}$.

Theorem 2 (Perfect matching associated to a collection of sets). *Let $\{S_i\}_{i \in I}$ be a collection of non-necessarily distinct subsets of an arbitrary set S.*

The directed bipartite digraph associated to $\{S_i\}_{i \in I}$ is built as the graph $G = \langle X, Y, E \rangle$ where $X = I$, $Y = \bigcup_{i \in I} S_i$ and $E = \{(i, x) \mid i \in I \text{ and } x \in S_i\}$.

Therefore, if R is an SDR of $\{S_i\}_{i \in I}$, then the subset of arcs $E' = \{(i, x) \mid i \in I \text{ and } x = R(i)\}$ is a perfect matching of G.

Preliminaries and definitions The Isabelle Archive of Formal Proofs contains a collection of theories regarding Graph Theory [25]. In particular, Noschinski and Neumann specified, in the theory *Digraph.thy*, the primary data structure *pre_digraph* as the basis to develop complex formalisations such as Kuratowski theorem and the existence of a Eulerian path on directed finite graphs. We also apply such a *record* to establish our formalisation.

```
record ('a,'b) pre_digraph =
  verts :: "'a set"          arcs :: "'b set"
  tail :: "'b ⇒ 'a"          head :: "'b ⇒ 'a"
```

Such a record from the theory mentioned above is used since the formalisation established in [25] contains specialised concepts intrinsic to the specific results formalised in it. For example, in the Isabelle AFP theory, *Kuratowski.thy* and *complete bipartite digraphs* are defined. However, there is no general specification of complete bipartite digraphs. Consequently, a small variety of basic concepts for graphs were specified. For instance, specifications of the *neighbourhood* of a vertex and the notion of *bipartite_digraph*, among others, are necessary to our development. In the following, some preliminary definitions are presented that were specified to establish the equivalence between the infinite versions of Hall's Theorem.

Arcs of a graph G have tails and heads in the set of vertices of the graph. The binary predicate `neighbour` on pairs of vertices u, v, holds if there exist and arc (u, v) or (v, u) in the graph. A `bipartite_digraph` is a *pre_digraph* G with two disjoint sets of vertices X and Y, whose union is the set of vertices of the graph, and such that all arcs in the graph have tails in X and heads in Y or vice versa.

```
definition tails:: "('a,'b) pre_digraph ⇒ 'a set" where
   "tails G ≡  { tail G e |e. e ∈ arcs G }"
```

```
definition tails_set :: "('a,'b) pre_digraph ⇒ 'b set ⇒ 'a set" where
  "tails_set G E ≡  { tail G e |e. e ∈ E ∧ E ⊆ arcs G }"

definition heads:: "('a,'b) pre_digraph ⇒ 'a set" where
  "heads G ≡  { head G e |e.  e ∈ arcs G }"

definition heads_set:: "('a,'b) pre_digraph ⇒ 'b set ⇒ 'a set" where
  "heads_set G E ≡  { head G e |e.  e ∈ E ∧ E ⊆ arcs G }"

definition neighbour::  "('a,'b) pre_digraph ⇒ 'a ⇒ 'a ⇒ bool" where
  "neighbour G v u  ≡
  ∃e. e∈ (arcs G) ∧ ((head G e = v ∧ tail G e = u) ∨
  (head G e  = u ∧ tail G e = v))"

definition neighbourhood:: "('a,'b) pre_digraph ⇒ 'a  ⇒ 'a set" where
  "neighbourhood G v ≡ {u |u. neighbour G u v}"

definition bipartite_digraph:: "('a,'b) pre_digraph ⇒ 'a set ⇒ 'a set
⇒ bool" where    "bipartite_digraph G X Y  ≡
                  (X ∪ Y = (verts G)) ∧  X ∩ Y = {} ∧
                  (∀e ∈ (arcs G).(tail G e) ∈ X ⟷ (head G e) ∈ Y)"
```

The specialised notion of directed bipartite digraphs used is specified in definition `dir_bipartite_digraph`. Such a graph is a bipartite digraph, consisting of a bi-partition of vertices X and Y in which all *arcs* have *tails* in the set X and *heads* in the set Y. Arcs with the same tail and head are equal.

```
definition dir_bipartite_digraph:: "('a,'b) pre_digraph ⇒ 'a set ⇒
'a set ⇒ bool"   where  "dir_bipartite_digraph G X Y  ≡
                   (bipartite_digraph G X Y) ∧  ((tails G = X) ∧
                   (∀e1 ∈ arcs G. ∀e2 ∈ arcs G. e1 = e2 ⟷
                    head G e1 = head G e2 ∧ tail G e1 = tail G e2))"
```

Definition `dirBD_matching` specifies a matching in a directed bipartite digraph G is specified as a subset E of the arcs of the graph, such that any pair of distinct arcs in E have neither the same head nor the same tail. A perfect matching, specified in definition `dirBD_perfect_matching`, is a matching in the digraph G that covers the set of vertices X.

```
definition dirBD_matching:: "('a,'b) pre_digraph ⇒ 'a set ⇒ 'a set ⇒
'b set ⇒ bool"  where   "dirBD_matching G X Y E  ≡
                    dir_bipartite_digraph G X Y ∧ (E ⊆ (arcs G)) ∧
                    (∀e1∈E. (∀e2∈ E.  e1 ≠ e2 ⟶
                    ((head G e1) ≠ (head G e2)) ∧
                    ((tail G e1) ≠ (tail G e2))))"

definition dirBD_perfect_matching::
 "('a,'b) pre_digraph ⇒ 'a set ⇒ 'a set ⇒ 'b set ⇒ bool"
```

```
  where   "dirBD_perfect_matching G X Y E ≡
                dirBD_matching G X Y E ∧ (tails_set G E = X)"
```

The theory `background_on_graphs` includes all definitions in this subsection. It specialised graphs according to the target formalisation requirements.

Building SDRs from perfect matchings Theorem 1, is specified as theorem `dir_BD_to_Hall` below. It uses the definition `E_head` that for any set of arcs E in a digraph and any vertex x, tail of some arc in E, selects the head, y, of an arc in E with tail x. The theorem states that for any directed bipartite digraph, $G = \langle X, Y, E\rangle$ with a perfect matching $E' \subseteq E$, the arcs of G, the family of sets given by the neighbourhoods of vertices $x \in X$ in G, $\{N(x)\}_{x\in X}$, the set of indices given by the set of vertices in X, and the representatives given by *E_head* using the perfect matching E', is an SDR. Since E' is a perfect matching, a unique arc with tail x in E' exists.

The required properties on the operator *E_head* on directed bipartite digraphs is that it gives an injective function over matchings also covering X over perfect matchings, which is stated as the crucial lemma `dirBD_matching_inj_on`. The proof requires proving a chain of auxiliary lemmas, including one stating the unicity of the operator *E_head* over matchings and then constructing an injective function that univocally maps tails into heads on the set of arcs E'.

Then, after unfolding definitions, one concludes that *(E_head G E)*, as an injective function on X, gives an SDR for the family of neighbourhoods of vertices in X, $\{N(X)\}_{x\in X}$, built from the graph G and the perfect matching E'.

```
definition E_head :: "('a,'b) pre_digraph ⇒ 'b set  ⇒ ('a  ⇒ 'a)"
  where  "E_head G E =
            (λx. (THE y. ∃ e. e ∈ E ∧ tail G e = x ∧  head G e = y))"

theorem dir_BD_to_Hall:
  "dirBD_perfect_matching G X Y E ⟶
  system_representatives (neighbourhood G) X (E_head G E)"
```

3.2 Formalising the graph-theoretical version of Hall's theorem

Here, we explain how the graph-theoretical version of Hall's theorem is obtained from its set-theoretical version formalised in [32]. The graph-theoretical version is stated as Theorem 3.

Theorem 3 (Hall - marriage-conditioned graph-theoretical version). *Let $G = \langle X, Y, E\rangle$ be a directed bipartite digraph. G contains a perfect matching covering the set of vertices X if and only if*

$$|J| \leq |N(J)| \quad \textit{for all} \quad J \subseteq X$$

As mentioned in the introduction, the theorem may be stated without paraphrasing the marriage condition to the context of graph theory.

Theorem 4 (Hall - graph-theoretical version). *Let $G = \langle X, Y, E\rangle$ be a directed bipartite digraph. G contains a perfect matching covering the set of vertices X if and only if for all finite $X_s \subset X$ the induced bipartite digraph has a perfect matching.*

These theorems are usually stated for finite graphs only. Also, in contrast to proofs presented in classical textbooks on (finite) graph theory (e.g., [38], [10]), their formalisations, given as the theorems *Hall_digraph*, at the end of this section, and *Hall_Graph*, in subsection 3.3, apply the combinatorial set-theoretical version of this theorem, obtained through application of the compactness theorem for propositional logic, extended for countable sets and published in [32].

The formalisation of Theorem 3 uses the Theorem 1 proved in Isabelle/HOL as described in Subsection 3.1 as theorem *dir_BD_to_Hall* and that states the correctness of the reduction of a directed bipartite digraph $G = \langle X, Y, E\rangle$ with a perfect matching E, to the family of neighbourhoods of vertices X, concluding that the operator *E_head* indeed builds an SDR from the perfect matching E.

The formalisation is based on applying two auxiliary lemmas relating the marriage condition for directed bipartite digraphs to perfect matchings.

The first auxiliary lemma, `marriage_necessary_graph`, states that if a directed bipartite graph has a perfect matching, then the marriage condition holds. Indeed, this lemma holds for arbitrary infinite graphs. Furthermore, relaxing the restriction on countable families to infinite families is possible since the lemma is proved as a consequence of the mechanisation of the fact that the existence of an SDR for arbitrarily infinite indexed families of finite sets implies the marriage condition. The last result was formalised through the theorem `marriage_necessity` part of the mechanisation reported in [32].

```
lemma marriage_necessary_graph:
assumes "(dirBD_perfect_matching G X Y E)"  and
        "∀i∈X. finite (neighbourhood G i)"
shows "∀J⊆X. finite J ⟶ (card J) ≤ card (⋃ (neighbourhood G ` J))"
```

Applying the transformation *(system_representatives (neighbourhood G) X (E_head G E))* through theorem *dir_BD_to_Hall* is the tricky part of this lemma. So, from the SDR, one obtains an injective function R from any subset J to their representatives in the union of neighbourhoods of elements $j \in J$ such that: $card\ J \leq card\ (\bigcup_{j \in J} N(j))$. The injectivity of R, guaranteed by theorem *dir_BD_to_Hall*, implies the desired inequation.

The second auxiliary lemma, `marriage_sufficiency_graph` below, states that if the marriage condition holds for a countable directed bipartite graph, then there exists a perfect matching.

```
lemma marriage_sufficiency_graph:
fixes G :: "('a, 'b) pre_digraph" and X:: "'a set"
assumes "dir_bipartite_digraph G X Y"  and
        "∀i∈X. finite (neighbourhood G i)"
and "∃g. enumeration (g:: nat ⇒ 'a)"
```

```
and "∃h. enumeration (h:: nat ⇒ 'b)"
shows
  "(∀J⊆X. finite J ⟶ (card J) ≤ card (⋃ (neighbourhood G ` J))) ⟶
  (∃E. dirBD_perfect_matching G X Y E)"
```

This lemma applies the formalisation of the countable set-theoretical version of Hall's theorem ([32]) to infer the existence of an SDR R for the countable indexed family of sets $\{N(i)\}_{i\in X}$. Applying the lemma is possible since the marriage condition for this family of sets is the premise of the target implication. From the system of representatives, it is possible to build the perfect matching as the set of arcs $\{(i, R(i))\}_{i\in X}$. Through two additional auxiliary lemmas, it is proved that this set covers the set of vertices X (lemma `perfect`) and is indeed a matching (lemma `dirBD_matching`). Therefore, one concludes that *(dirBD_perfect_matching G X Y* $\{(i, R(i))\}_{i\in X}$*)*.

Finally, the countable graph-theoretical version of Hall's theorem (Theorem 3), specified as theorem `Hall_digraph`, is formalised as below. The use of necessity and sufficiency auxiliary lemmas is highlighted in the mechanisation.

```
theorem Hall_digraph:
fixes  G :: "('a, 'b) pre_digraph" and X:: "'a set"
assumes "dir_bipartite_digraph G X Y"
and "∀i∈X. finite (neighbourhood G i)"
and  "∃g. enumeration (g:: nat ⇒ 'a)"
and  "∃h. enumeration (h:: nat ⇒ 'b)"
shows "(∃E. dirBD_perfect_matching G X Y E) ⟷
        (∀J⊆X. finite J ⟶ card J ≤ card (⋃(neighbourhood G ` J)))"
proof
 assume hip1: " ∃E. dirBD_perfect_matching G X Y E"
 show"∀J⊆X. finite J ⟶ card J ≤ card ⋃ (neighbourhood G ` J)"
    using hip1 assms(1-2) marriage_necessary_graph[of G X Y] by auto
next
 assume hip2 :"∀J⊆X.finite J ⟶ card J ≤card⋃(neighbourhood G ` J)"
 show"∃E.dirBD_perfect_matching G X Y E"
    using assms  marriage_sufficiency_graph[of G X Y] hip2
  proof-
   have"∀J⊆ X.finite J⟶ card J ≤ card ⋃(neighbourhood G ` J)
          ⟶   (∃E. dirBD_perfect_matching G X Y E)"
      using assms marriage_sufficiency_graph[of G X Y] by auto
    thus ?thesis using hip2 by auto
  qed
qed
```

3.3 Alternative formalisation using Isabelle *locales*

Locales are an extension of Isabelle (Isar) that provide support for modular reasoning allowing dependent typing in a straight forward manner. Locales were initially developed by Kammüller [22] to support reasoning in abstract algebra, but are applied in a variety of domains [6,7]. This section discusses the formalisation of the difficult direction of the graph version of Hall's theorem directly over graph notions, as stated by Theorem 4, using locales.

Initially, locales are used to specify an indexed family of sets from which the notion of SDR is specified providing an injective function *repr* from the set of indices I to distinct elements in each set of the family.

```
locale set_family =
  fixes I :: "'a set" and X :: "'a ⇒ 'b set"

locale sdr = set_family +
  fixes repr :: "'a ⇒ 'b"
  assumes  inj_repr: "inj_on repr I" and
           repr_X: "x ∈ I ⟹ repr x ∈ X x"
```

Then, the notions of bipartite digraph and countable bipartite digraph $G = \langle X, Y, E\rangle$ with finite sets of neighbourhoods for each vertex $x \in X$ are defined as below.

```
locale bipartite_digraph =
  fixes X :: "'a set" and Y :: "'b set" and E :: "('a × 'b) set"
  assumes E_subset: "E ⊆ X × Y"

locale Count_Nbhdfin_bipartite_digraph =
  fixes X :: "'a:: countable set" and Y :: "'b:: countable set"
           and E :: "('a × 'b) set"
  assumes E_subset: "E ⊆ X × Y"
  assumes Nbhd_Tail_finite: "∀x ∈ X. finite {y. (x, y) ∈ E}"
```

In the sequel, matching over bipartite digraphs and perfect matching are specified using locales. The succinctness of locales is observed clearly in the definition of perfect matching. For this, it is only required to add to the notion of matching the assumption that the matching covers the set of vertices X.

```
locale matching = bipartite_digraph +
  fixes M :: "('a × 'b) set"
  assumes M_subset: "M ⊆ E"
  assumes M_right_unique: "(x, y) ∈ M ⟹ (x, y') ∈ M ⟹ y = y'"
  assumes M_left_unique: "(x, y) ∈ M ⟹ (x', y) ∈ M ⟹ x = x'"

locale perfect_matching = matching +
  assumes M_perfect: "fst ` M = X"
```

Then, using the locales for systems of distinct representatives, *sdr*, and for perfect matchings, respectively, two lemmas can be easily established, proving how a perfect matching can be built from an SDR, and how a perfect matching gives rise to an SDR. The former lemma uses the injective function *repr* in the locale for *sdr*, building the perfect matching as the set of edges $\{(x, \mathit{repr}\ x) \mid x \in I\}$. The latter lemma uses the set $M \subseteq E$ in the locale for perfect matching to build the SDR using as a set of indices the vertices X, as the family of indexed sets the function mapping indices into their finite neighbourhoods, $x \in X$, $\lambda x.\{y \mid (x,y) \in E\}$, and as injective function $\lambda x.\{y \mid (x,y) \in M\}$.

```
lemma (in sdr) perfect_matching:
  "perfect_matching I (⋃i∈I. X i) (Sigma I X) {(x, repr x)|x. x ∈ I}"
```

```
lemma (in perfect_matching) sdr:
  "sdr X (λx. {y. (x,y) ∈ E}) (λx. the_elem {y. (x,y) ∈ M})"
```

Finally, the difficult direction of Hall's theorem, as stated by Theorem 4, for countable infinite graphs is specified using the locale for countable bipartite digraphs with finite neighbourhoods for all vertices $x \in X$. The theorem below formalises that if the subgraph induced by any finite subset X_s of X has a perfect matching, then the whole graph has a perfect matching.

```
theorem (in Count_Nbhdfin_bipartite_digraph) Hall_Graph:
  shows "(∀ Xs ⊆ X. (finite Xs) ⟶
           (∃ Ms.   perfect_matching Xs
                                       {y. x ∈ Xs  ∧ (x,y) ∈ E}
                                       {(x,y). x  ∈ Xs  ∧ (x,y) ∈ E}
                                       Ms))
        ⟶ (∃ M.  perfect_matching X Y E M)"
```

The proof uses the hypotheses of the existence of a perfect matching, M_S, for each bipartite digraph induced by any finite $X_S \subset X$. Using the previous lemma *(in perfect_matching) sdr*, it is possible to construct an SDR for the associated family of sets of neighbourhoods of vertices incident to vertices in X_S. Then, the existence of different images of the injective function to the distinct representatives, *repr* in the locales for *sdr*, permits inferring that $|X_S| \leq \cup_{x \in X_s}\{y \mid (x,y) \in E\}$. Notice that this condition corresponds to the set-theoretical marriage condition. Thus, applying the set-theoretical version of Hall's theorem one concludes that the whole digraph has an SDR. Finally, the existence of a perfect matching for the whole digraph is concluded by applying the previous lemma *(in sdr) perfect_matching*.

Notice that the other direction of Theorem 4 is easy; indeed, the restriction of the perfect matching of the whole graph to the subgraph induced by any subset $X_S \subset X$ is a perfect matching of the induced subgraph.

4 Related Work

4.1 Automation versus interactive comprehensive proofs

As mentioned in the abstract, our primary interest in developing such a detailed formalisation is to provide insight to Mathematicians and Computer Scientists about the usefulness of proof assistants. So, the high granularity used in presenting definitions and proof steps is essential. Using the Isabelle Sledgehammer [26,37] the user may infer proofs without having a clear idea of how these proofs are obtained, which is not our objective. To summarise the steps inferred by the Sledgehammer, it is recommended to restrict it to *isar* proofs. Such an alternative approach, oriented towards automation, is presented at the end of the formalisation using locales [6,7].

In synthesis, our educational goal prioritises the application of proof assistants as *interactive theorem provers* and not as *automated theorem provers*. This is the spirit we have followed teaching for years computer science and Math students in our institutions as reported in [4] (on the adequate application of interactive theorem provers to motivate mathematicians), [3] (on the application of the proof assistant PVS to teach computer science, mathematicians, and engineering students to verify algorithms), and in [5] (on teaching computational logic to computer science, engineering and mathematics students, illustrating the application of the Gentzen's sequent-style calculus implemented in the proof assistant PVS).

4.2 On Hall's theorem and other combinatorial theorems

Extensions to the infinite case from theorems equivalent to Hall's marriage theorem in the finite case are generally not straightforward. In addition to the infinite version of Hall's marriage theorem, our development includes formalisations of infinite versions of De Bruijn-Erdös graph colouring theorem ([11]) and König lemma ([23]), obtained from the compactness theorem for predicate logic (theorems available through the links `k_coloring` and `Koenig_Lemma`, respectively). Moreover, even such extensible theorems would not necessarily be proven by the compactness theorem and elementary techniques. An example is König's duality theorem, proved by Aharoni [1], and subsequently studied in detail by Aharoni et al. [2]. This theorem states that in every bipartite graph $G = \langle X, Y, E \rangle$, *there exists* a matching $M \subseteq E$ such that selecting one vertex from each arc in M one has a cover of the graph. König duality theorem is a strong form of the finite, well-known König-Egerváry theorem that states that in a finite bipartite graph, the size of a maximal matching is equal to the size of a minimal cover [24]. The vital difference of the duality theorem is that such a cover of the graph cannot be extracted from an arbitrary matching. Indeed, from a matching, it is possible to build a cover of the same cardinality as the cardinality of the matching, but not that it covers the graph. So, the notion of *König cover* came to arise, which is defined as a cover of the graph that consists of a selection of one vertex from each arc of a matching.

Lifting results from the finite to the infinite through the application of compactness (of König's lemma) corresponds to a recursive construction of a procedure that produces the target solution in the degree of unsolvability of the halting problem [2]. Such a recursive construction is possible for Dilworth's theorem (restricting the maximal anti-chains in infinite partial ordered sets to be finite - [12], see also Sec. 2.5 in [19]) but not for König's duality theorem. Indeed, Aharoni et al. [2] proved that the complexity of constructing covers exceeds the complexity of the halting problem; it is even a problem of higher complexity than answering all first-order questions about arithmetic. Also, they proved that the compactness theorem and König's lemma do not suffice to prove the duality theorem and other related results in matching theory.

The first formalisation of the finite version of Hall's Theorem was developed in Mizar by Romanowicz and Grabowski [29]. Also, there are formalisations in Isabelle/HOL by Jiang and Nipkow [21]. These formalisations follow Rado's proof [27], but the last one also includes a mechanisation based on Halmos and Vaughan's proof [18]. In addition, Coq has a formalisation that uses formalisations of Dilworth's decomposition theorem and bi-partitions in graphs [34]. An earlier formalisation of Dilworth's theorem in Mizar is presented in [30]. Recently, Gusakov, Mehta and Miller [16] presented three different proofs of the finite version of Hall's theorem in Lean in terms of indexed families of finite subsets, of the existence of injections that saturate binary relations over finite sets and of matchings in bipartite graphs. Related combinatorial results are reported in recent works by Doczkal et al. in their graph theory Coq library (e.g., [13], [15], and [14]). Additionally, Singh and Natarajan formalised in Coq other combinatorial results as the perfect graph theorem and a weak version of this theorem (e.g., [35], [36]).

Known mechanisations of the enumerable version of the set-theoretical version of Hall's theorem appear in the formalisation used in the authors' work, previously discussed, [32], and in Gusakov, Mehta, and Miller's work [16]. The former work uses the compactness theorem for predicate logic. In the latter work, the authors apply an *inverse limit* version of the König's lemma. This lemma states that if $\{X_i\}_{i\in\mathbb{N}}$ is an indexed family of nonempty finite sets with functions $f_i : X_{i+1} \to X_i$, for each $i \in \mathbb{N}$, then there exists a family of elements $x \in \prod_i X_i$ such that $x_i = f_i(x_{i+1})$, for all $i \in \mathbb{N}$. König's lemma follows from this infinite limit version by choosing as set X_i the paths of length i from the root vertex v_0 in a tree. So, the function f_i maps paths in X_{i+1} into the paths without their last arc that belong to X_i. The inverse limit consists of the infinite chain of functions $f_1, f_2, \ldots$. König's lemma is applied to prove the enumerable version of Hall's theorem by taking M_n as the set of all matchings on the first n indices of I (i.e., the set of all possible SDRs for the sets $S_1, \ldots, S_n$), and $f_n : M_{n+1} \to M_n$ as the restriction of a matching to a smaller set of indices. Since the marriage condition holds for the finite indexed families, each M_n is nonempty, and by König's lemma, an element of the inverse limit gives a matching on I.

5 Conclusions and Future Work

This paper presented two formalisations in Isabelle/HOL of the graph-theoretical version of Hall's theorem for countable (infinite) graphs. The prominent feature of the first formalisation is following a presentation close to pen-and-paper proofs but dissecting all minimal required steps in the assisted proof. Exhibiting minimal details, usually omitted in practice, is relevant to highlight to Math and CS students and professionals the relevance of mechanised proofs. On the other hand, the second one is more succinct and uses Locales, which are powerful mechanisms to deal with parametric theories in Isabelle/HOL.

These developments will enable other mechanisations of infinite combinatorial, set-theoretical, and graph-theoretical results related to the compactness theorem for predicate logic and its derivations, such as König lemma, Hall's marriage theorem, and de Bruijn-Erdös k-colouring theorem, as well as generalisations of Dilwort's theorem.

An exciting challenge for future research consists in developing the required formal background in proof assistants to enable the formalisation of other theorems which do not extend straightforwardly from the results mentioned above, such as the König duality theorem, among others.

Acknowledgements We want to thank Cezary Kaliszyk, René Thiemann, Fabian Huch, and Yutaka Nagashima, who kindly shared their expertise on Isabelle/HOL.

References

1. Ron Aharoni. König's Duality Theorem for Infinite Bipartite Graphs. *Journal of the London Mathematical Society*, s2-29(1), 1984. https://doi.org/10.1112/jlms/s2-29.1.1.
2. Ron Aharoni, Menachem Magidor, and Richard A Shore. On the Strength of König's Duality Theorem for Infinite Bipartite Graphs. *Journal of Combinatorial Theory Series B*, 54(2):257–290, 1992. https://doi.org/10.1016/0095-8956(92)90057-5.
3. Ariane Alves Almeida, Ana Cristina Rocha Oliveira, Thiago Mendonça Ferreira Ramos, Flávio L. C. de Moura, and Mauricio Ayala-Rincón. The computational relevance of formal logic through formal proofs. In *Proc. Formal Methods Teaching - Third International Workshop and Tutorial, FMTea*, volume 11758 of *Lecture Notes in Computer Science*, pages 81–96. Springer, 2019. https://doi.org/10.1007/978-3-030-32441-4_6.
4. Mauricio Ayala-Rincón and Thaynara Arielly de Lima. Teaching interactive proofs to mathematicians. In *Proc. 9th International Workshop on Theorem Proving Components for Educational Software, ThEdu*, volume 328 of *EPTCS*, pages 1–17, 2020. https://doi.org/10.4204/EPTCS.328.1.
5. Mauricio Ayala-Rincón and Flávio L. C. de Moura. *Applied Logic for Computer Scientists - Computational Deduction and Formal Proofs*. Undergraduate Topics in Computer Science. Springer, 2017. https://doi.org/10.1007/978-3-319-51653-0.
6. Clemens Ballarin. Locales and Locale Expressions in Isabelle/Isar. In Stefano Berardi, Mario Coppo, and Ferruccio Damiani, editors, *Types for Proofs and Programs, International Workshop, TYPES 2003, Torino, Italy, April 30 - May 4, 2003, Revised Selected Papers*, volume 3085 of *Lecture Notes in Computer Science*, pages 34–50. Springer, 2003. https://doi.org/10.1007/978-3-540-24849-1_3.
7. Clemens Ballarin. Locales: A Module System for Mathematical Theories. *J. Autom. Reason.*, 52(2):123–153, 2014. https://doi.org/10.1007/s10817-013-9284-7.
8. Robert D. Borgersen. Equivalence of seven major theorems in combinatorics, 2004. Talk available at Department of Mathematics, University of Manitoba, Canada. https://home.cc.umanitoba.ca/~borgerse/Presentations/GS-05R-1.pdf.
9. Peter J. Cameron. *Combinatorics: Topics, Techniques, Algorithms*. Cambridge University Press, 1994.

10. Gary Chartrand, Linda Lesniak, and Ping Zhang. *Graphs and Digraphs*. Chapman & Hall/CRC, 5th edition, 2010.
11. Nicolaas Govert De Bruijn and Pál Erdös. A Colour Problem for Infinite Graphs and a Problem in the Theory of Relations. *Indagationes Mathematicae (Proceedings)*, 54:371–373, 1951.
12. Robert P. Dilworth. A Decomposition Theorem for Partially Ordered Sets. *Annals of Mathematics*, 51(1):161–166, 1950. https://doi.org/10.2307/1969503.
13. Christian Doczkal, Guillaume Combette, and Damien Pous. A Formal Proof of the Minor-Exclusion Property for Treewidth-Two Graphs. In *Proceedings 9th International Conference on Interactive Theorem Proving - ITP*, volume 10895 of *Lecture Notes in Computer Science*, pages 178–195. Springer, 2018. https://doi.org/10.1007/978-3-319-94821-8_11.
14. Christian Doczkal and Damien Pous. Completeness of an axiomatization of graph isomorphism via graph rewriting in Coq. In *Proceedings of the 9th ACM SIGPLAN International Conference on Certified Programs and Proofs - CPP*, pages 325–337. ACM, 2020.
15. Christian Doczkal and Damien Pous. Graph Theory in Coq: Minors, Treewidth, and Isomorphisms. *Journal of Automated Reasoning*, 64(5):795–825, 2020. https://doi.org/10.1007/s10817-020-09543-2.
16. Alena Gusakov, Bhavik Mehta, and Kyle A. Miller. Formalizing Hall's Marriage Theorem in Lean. *arXiv abs/2101.00127[math.CO]*, 2021. https://doi.org/10.48550/arxiv.2101.00127.
17. Philip Hall. On representatives of subsets. *London Mathematical Society*, 10:26–30, 1935. https://doi.org/10.1112/jlms/s1-10.37.26.
18. Paul R. Halmos and Herbert E. Vaughan. The Marriage Problem. *American Journal of Mathematics*, 72(1):214–215, 1950.
19. Egbert Harzheim. *Ordered Sets*, volume 7 of *Advances in Mathematics*, chapter 2. General relations between posets and their chains and antichains. Springer, 2005.
20. Dongchen Jiang and Tobias Nipkow. Hall's Marriage Theorem. *Archive of Formal Proofs*, 2010, 2010.
21. Dongchen Jiang and Tobias Nipkow. Proof Pearl: The Marriage Theorem. In *Proceedings First International Conference on Certified Programs and Proofs - CPP*, volume 7086 of *Lecture Notes in Computer Science*, pages 394–399, 2011. https://doi.org/10.1007/978-3-642-25379-9_28.
22. Florian Kammüller. Modular Reasoning in Isabelle. In *Proceedings Automated Deduction - CADE-17, 17th International Conference on Automated Deduction*, volume 1831 of *Lecture Notes in Computer Science*, pages 99–114. Springer, 2000.
23. Dénes König. Über eine Schlussweise aus dem Endlichen ins Unendliche. *Acta Sci. Math. (Szeged)*, 3(2-3):121–130, 1927.
24. Dénes König. *Theorie Der Endlichen und Unendlichen Graphen: Kombinatorische Topologie Der Streckenkomplexe*, volume 16 of *Mathematik und ihre Anwendungen in Monographien und Lehrbüchern*. Chelsea, 1936.
25. Lars Noschinski. Graph theory. *Archive of Formal Proofs*, April 2013. http://isa-afp.org/entries/Graph_Theory.html, Formal proof development.
26. Lawrence C. Paulson and Jasmin Christian Blanchette. Three years of experience with sledgehammer, a practical link between automatic and interactive theorem provers. In *The 8th International Workshop on the Implementation of Logics, IWIL 2010, Yogyakarta, Indonesia, October 9, 2011*, volume 2 of *EPiC Series in Computing*, pages 1–11. EasyChair, 2010. https://doi.org/10.29007/36dt.

27. Richard Rado. Note on the transfinite case of Hall's theorem on representatives. *London Mathematical Society*, S1-42(1):321–324, 1967. `https://doi.org/10.1112/jlms/s1-42.1.321`.
28. Philip F. Reichmeider. *The equivalence of some combinatorial matching theorems*. Polygonal Publishing House, 1985.
29. Ewa Romanowicz and Adam Grabowski. The Hall Marriage Theorem. *Formalized Mathematics (University of Białystok)*, 12(3):315–320, 2004. `https://fm.mizar.org/2004-12/pdf12-3/hallmar1.pdf`.
30. Piotr Rudnicki. Dilworth's Decomposition Theorem for Posets. *Formalized Mathematics*, 17(4):223–232, 2009. `https://doi.org/10.2478/v10037-009-0028-4`.
31. Fabián Fernando Serrano Suárez. *Formalización en Isar de la Meta-Lógica de Primer Orden*. PhD thesis, Departamento de Ciencias de la Computación e Inteligencia Artificial, Universidad de Sevilla, Spain, 2012. `https://idus.us.es/handle/11441/57780`. In Spanish.
32. Fabián Fernando Serrano Suárez, Mauricio Ayala-Rincón, and Thaynara Arielly de Lima. Hall's Theorem for Enumerable Families of Finite Sets. In *Proc. Int. Conf. on Intelligent Computer Mathematics CICM*, volume 13467 of *Lecture Notes in Computer Science*, pages 107–121. Springer, 2022. `http://doi.org/10.1007/978-3-031-16681-5_7`.
33. Fabián Fernando Serrano Suárez, Mauricio Ayala-Rincón, and Thaynara Arielly de Lima. Mechanising Hall's Theorem for Countable Graphs. EasyChair Preprint no. 10365, 2023. Easy Chair Preprint of talk presented at LPAR 2023. `https://easychair.org/publications/preprint/g3F7`.
34. Abhishek Kr Singh. Formalization of some central theorems in combinatorics of finite sets. *arXiv abs/1703.10977[cs.Lo]*, 2017. Short presentation at the 21st International Conference on Logic for Programming, Artificial Intelligence and Reasoning - LPAR. `https://doi.org/10.48550/arxiv.1703.10977`.
35. Abhishek Kr Singh and Raja Natarajan. Towards a Constructive Formalization of Perfect Graph Theorems. In *Proceedings 8th Indian Conference on Logic and Its Applications - ICLA*, volume 11600 of *Lecture Notes in Computer Science*, pages 183–194. Springer, 2019. `https://doi.org/10.1007/978-3-662-58771-3_17`.
36. Abhishek Kr Singh and Raja Natarajan. A constructive formalization of the weak perfect graph theorem. In *Proceedings of the 9th ACM SIGPLAN International Conference on Certified Programs and Proofs - CPP*, pages 313–324. ACM, 2020. `https://doi.org/10.1145/3372885.3373819`.
37. Nik Sultana, Jasmin Christian Blanchette, and Lawrence C. Paulson. LEO-II and satallax on the sledgehammer test bench. *J. Appl. Log.*, 11(1):91–102, 2013. `https://doi.org/10.1016/j.jal.2012.12.002`.
38. Douglas Brent West. *Introduction to Graph Theory*. Pearson Modern Classics for Advanced Mathematics. Pearson Education, Inc, second edition, 2001.